


Application Note		作成 100902 更新 120209	資料番号 an2k-015
テーマ	ステップモータオルゴール (MPC-1000 デモ機)		
使用機器	MPC-1000、XY-03、タッチパネル、USB メモリ、温湿度計 他		

## ■機器構成



## ■概要

- ・ 少数 I/O とステップモータで構成された小型卓上機を想定したデモ機です。
- ・ ユーザーインターフェースはタッチパネルです。
- ・ パルスモータテーブルのピッチ送り(位置制御)、定速回転、USB メモリからファイルを読み込んで動作変更、などを行います。
- ・ USB メモリに保存されている点データのフォーマットに従い PGB のパルスレートを変えて音階を作り、オルゴールのようにメロディーを演奏します。
- ・ タッチパネルの鍵盤操作を USB メモリに保存、再生します。再生スピードは VR(AD 入力)で調整します。
- ・ タッチパネルの時計で MPC-1000 の時計をセットします。
- ・ RS-485 でマルチドロップ接続した温湿度計のデータを USB メモリに CSV 形式で記録します。

## ■特徴


### MPC-1000

- ・ 入力 16 点(そのうち 7 点は AD と兼用、パルス使用時は 1 軸につき 2 点が PG 制御用となります)
- ・ 出力 16 点(パルス使用時は 1 軸につき 2 点がパルスポートになります。差動パルス出力と排他使用。)
- ・ RS-232C 3ch(プログラム 1、ユーザー 2、そのうち 1ch は RS-485 と排他使用)。パナソニック MEWNET-COM、オムロン CompoWay/F 準拠プロトコル搭載。
- ・ USB メモリを利用してデータロガーなどにも応用できます。
- ・ 言語仕様、拡張ボード・ラック、プログラム開発環境は MPC-2000 と共通です。
- ・ 詳細はパンフレット、マニュアルでご覧下さい。

### Quattro-2

- ・ Quattro-2 は 4 軸の 2 相マイクロステップモータドライバですが、本機ではそのうち 2 軸を異なる設定で使っています。設定は専用ソフトで行います。  
X 軸(PGA)：励磁モード 1/4 分割、ドライブ電流 0.6A、スリープ電流 0.1A  
U 軸(PGB)：励磁モード 1/2 分割、ドライブ電流 0.8A、スリープ電流 0.4A


■ 結線



## ■タッチパネル部品アドレス


※実際の表示とは異なる場合があります。ページ表示順番はプログラムに依存します。

・&H63 ページ



入出力部品無し

・&H80 ページ



DT993 [メッセージ部品] 文字設定

- 0 : 空白
- 1 : ファイル名取得中
- 2 : 点データ読み込み中
- 3 : 読み込みました
- 4 : 点データではありません
- 5 : ファイルを選んでください
- 6 : 動作中


DT994 [データ部品] ASCII 12 文字表示

DT200,DT210 [データ部品] DEC(2W) 5 桁表示

DT220 [データ部品] DEC(2W 符号なし)6 桁表示


R900,901,903 [スイッチ部品] モーメンタリ

・ &H81 ページ




DT101 [データ部品] DEC(1W 符号なし)3桁表示  
 DT102 [データ部品] DEC(1W 符号なし)5桁表示  
 DT140 [データ部品] ASCII 12文字表示  
 R200~204 [スイッチ部品] モーメンタリ

・ &H82 ページ




DT140 [データ部品] ASCII 12文字表示  
 R0 [スイッチ部品] オルタネイト

・ &H85 ページ




R310 [スイッチ部品] オルタネイト  
 DT124 [データ部品] ASCII 10文字表示  
 DT116 [データ部品] ASCII 8文字表示  
 DT135, DT140, DT145, DT150 [データ部品] ASCII 4文字表示  
 DT160, DT161, DT163 [データ部品] DEC(1W 符号なし)2桁表示  
 DT162 [データ部品] DEC(1W 符号なし)4桁表示  
 DT230 [データ部品] DEC(1W 符号なし)4桁表示



R300~307 [スイッチ部品] モーメンタリ  
 R400~407 [ランプ部品]  
 R310 [スイッチ部品] オルタネイト  
 R311 [スイッチ部品] モーメンタリ  
 DT230 [データ部品] DEC(1W 符号なし) 4桁表示

### ■タッチパネル通信設定

タッチパネル通信はプログラム読み込み時などに停止します。そのとき GT は通信エラーになり、画面右上に「ERFF」と表示されます。プログラムが走り通信開始を再開すると復帰しますが、時間がかかることがあります。通信中断時間の短縮策として 本体環境設定 の 通信エラー時処理 を調整みてください。



## ■MPC サンプルプログラム

```
*****
/* MPC-1000 デモ機 プログラム
/* 100120
/* MPC-1000(SH7030) BL/I 1.12_18 2010/01/15
*****
/* 接続機器
/* PGA, PGB : 2相ステップモータ。マイクロステップ駆動。
/* タッチパネル : GT11 RS-232C 38400bps MEWNET
/* USBメモリ : 点データ格納
/* 入力 : SW(194)=緑スイッチ、SW(195)=黄スイッチ
/*         SW(196)=赤スイッチ、SW(197)=AUTO/MANU
/*         SW(198)=PGA org、IN(25)(SW(200)~SW(203))=DSW
/*         SW(204)=PGA ready、SW(206)=PGB ready
/*         ※このデモプログラムは緑、黄、赤スイッチは
/*         START, STOP, RESET として使用していません。
/*         ※SW(192), SW(193)はA/D用にリザーブ(このデモでは未使用)
/* AD入力 : 可変抵抗器。内部5V-GND使用。
/* 出力 : ON 0=緑LED、ON 1=黄LED、ON 2=赤LED、ON 3=ブザー
/*         ON 4~ON 7=分配基盤OUT1~4(このデモでは未使用)
/* RS-485 : CHINO 温湿度計
*****
/* QUATTRO Parameters
/* X_AXIS=1/4 Drv=0.6A Slp=0.2A Fall=0.1sec
/* Y_AXIS=1/4 Drv=0.0A Slp=0.0A Fall=0.0sec
/* U_AXIS=1/2 Drv=0.8A Slp=0.4A Fall=1.0sec
/* Z_AXIS=1/16 Drv=0.0A Slp=0.0A Fall=0.0sec
/* PULSE = CW /CCW Sync Disabled
/* temp Limit =45deg
/* ※このデモで使うのはX_AXIS、U_AXIS
*****
/* 点データ書式 (例 ..¥usb_mem¥doremi.p2k)
/*      P No  PGB Dir  KEY ARRAY No  Delay  PGA Dir
/* SETP 1001  1      20      1000  1
/* SETP 1002  1      30      300   1
/*   ・ P No : 点番号 1001 から使用
/*   ・ PGB Dir : PGB の回転方向 1=CW, -1=CCW
/*   ・ KEY ARRAY No : KEY 配列の番号。KEY 配列に周波数が書いてある。
/*   ・ Delay : 回転(または休み)時間
/*   ・ PGA Dir : PGA の回転方向。1=CW、0=CCW
*****
/* 動作概要
/* AUTO モード
/*   ・ PGA(スライドテーブル)とPGB(回転)は別タスクで制御。
/*   ・ PGBタスクがタッチパネルでUSBメモリファイルを選択、点データを読み込む。
/*   ・ 点データの書式に従ってモータを回転、メモリI/Oを操作する。
/*   ・ PGAとPGBのスタート・折り返しはメモリI/Oで同期をとっている。
/* MANU モード
/*   ・ DSWで動作切り替え
/*   1: 鍵盤(録音、再生)、AD入力
/*   2: PGA, PGB ジョグ移動
/*   3: PGA 単独動作
/*   4: PGB 単独動作
/*   5: PGA, PGB 同期動作
/*   それ以外: 時計表示
*****
/* システムで使用するタスク(ユーザープログラムで使用不可)
/* TASK 31 : タッチパネル通信
/* TASK 29 : USBメモリアクセス
*****
```

```

/* 点データ使用エリア
/* P(1000) ~ 音楽演奏(USB MEMORY READ)
/* P(2000) ~ (2099) 鍵盤操作記録(USB MEMORY WRITE/READ)
/* P(3000) ~ 温湿度記録

```

```

DIM KEY(110)          /* 音階テーブル

```

```

KEY(10)=0            /* 休み
KEY(20)=900         /* ド
KEY(25)=1000-60     /* レ♭
KEY(30)=1000        /* レ
KEY(40)=1105        /* ミ
KEY(50)=1190        /* ファ
KEY(60)=1350        /* ソ
KEY(70)=1500        /* ラ
KEY(75)=1680-60     /* シ♭
KEY(80)=1680        /* シ
KEY(90)=1800        /* ド
KEY(100)=2000       /*

```

```

`DEBUG=1            /* デバッグモードはFTMW読み込みの点データを使う

```

```

ON_USB
TIME 500

```

```

MEWNET 38400 2      /* タッチパネル ボーレート、通信ポート
S_MBK &H63 8       /* タッチパネル オープニング画面

```

```

CNFG# 1 RS485 "9600b7pes1NONE"

```

```

CONST PGA_ORG 198   /* PGA 原点LS
CONST PGA_RDY 204   /* PGA READY
`CONST PGB_ORG 199
CONST PGB_RDY 206

```

```

SETIO
GOSUB *SET_PG_ACCEL /* 加減速テーブル作成
GOSUB *CLOCK_SET    /* タッチパネルの時計でMPCの時計をアジャスト

```

```

FORK 1 *MODE
END          /* メインタスク終了

```

```

*MODE          /* AUTO/MANU モード切替
DO

```

```

IF SW(197)==0 THEN /* AUTO
GOSUB *JOB_STOP
FORK 2 *AUTO_PGA
FORK 3 *AUTO_PLAY
WAIT SW(197)==1    /* MODE SW
ELSE                /* MANU
DO
GOSUB *JOB_STOP
DSW=IN(25)&&HF      /* 全面パネル DSW
SELECT_CASE DSW
CASE 1 : FORK 2 *PGB_DOREMI
        FORK 3 *AD
CASE 2 : FORK 2 *PG_TEST
CASE 3 : FORK 2 *PGA_TEST
CASE 4 : FORK 2 *PGB_TEST
CASE 5 : FORK 2 *PGA_TEST
        FORK 3 *PGB_TEST
CASE_ELSE : FORK 2 *CLOCK_TEMP_DISP

```

```

                FORK 3 *AD
END_SELECT
WAIT IN (25) &&HF<>DSW OR SW(197)==0 /* DSWが変わるか MODEが変わるか
IF SW(197)==0 THEN
    BREAK /* EXIT FROM MANU
END_IF
LOOP
END_IF
TIME 100 /* MODE SW チャタ防止 (このSWはチャタる)

LOOP

*JOB_STOP /* タスク、パルス停止、出力初期化
QUIT 2 3
TIME 50
OFF PGA PGB /* PGA PGB DEACTIVATE モータ回転停止
OFF 0 1 2
OFF -1 -2
PULSE_OUT VOID /* パルスアウト解除
TIME 100 /* WAIT FOR PG DEACTIVATE
RETURN

*AUTO_PGA /* PGA 駆動
OFF -1 -2

DO

    WAIT SW(-1)==1 /* PLAYタスクからのスタート待ち
    GOSUB *PGA_HOME
    ON -2 /* PLAYタスクへHOME完了フラグ

    GOSUB *AUTO_PGA_POS_DISP
    FOR I_=1 TO 11 /* ピッチ送り
        PR I_
        PGA "R" 1000 /* 相対移動
        WAIT SW(PGA_RDY)==1
        GOSUB *AUTO_PGA_POS_DISP
        IF SW(-1)==0 THEN
            BREAK
        END_IF
    NEXT I_

    WAIT SW(-1)==0

    PGA "F" 2
    PGA "M" 0 /* 座標0へ絶対移動
    WAIT SW(PGA_RDY)==1
    GOSUB *AUTO_PGA_POS_DISP

    WAIT SW(-2)==0

LOOP

*AUTO_PGA_POS_DISP /* PGA 座標表示
/* MPC-1000のPGA、PGBの座標はパルス発生中には取得できない
PGA "C"
S_MBK V_PGA 200~Lng /* タッチパネル表示(ロングワード)
RETURN

*AUTO_PLAY /* PGB 演奏

TIME 100
S_MBK &H80 8 /* タッチパネル 画面切り替え

```



```

S_MBK 0 200~Lng          /* タッチパネル表示(ロングワード)
S_MBK 0 210~Lng          /* タッチパネル表示(ロングワード)

ON_USB

IF DEBUG==0 THEN
  GOSUB *USB_READ        /* USBメモリからファイル名を取得
END_IF

DO

IF DEBUG==0 THEN
  GOSUB *USB_MEM         /* ファイル選択・読み込み・動作開始
  S_MBK 6 993           /* タッチパネル“動作中”
ELSE
  DO
    PRINT "DEBUG MODE"
    ON 0
    WAIT SW(194)==1
    OFF 0
    IF X(1001)==0 THEN /* データが無い
      PRINT "DATA NOE"
    ELSE
      BREAK
    END_IF
  LOOP
END_IF

ON -1                    /* PGA 移動開始合図
WAIT SW(-2)==1          /* PGA軸 HOME完了待ち
I_=1001
J_=0
DO

  IF X(I_)<>0 THEN

    TONE=X(I_)*KEY(Y(I_)) /* X:DIR, Y:TONE TABLE
    S_MBK ABS(TONE) 210~Lng /* タッチパネルに周波数を表示
    PGB "G" TONE          /* 定速回転
    ON J_                 /* 緑・黄・赤LED点滅(意味は無い)
    TIME U(I_)           /* 回転(休止)時間
    IF Z(I_)=0 THEN
      OFF -1             /* PGA 反転合図
    END_IF
    OFF J_
    J_=J_+1 : IF J_>2 THEN : J_=0 : END_IF

  END_IF

  I_=I_+1
  IF X(I_)==999 THEN     /* END OF DATA
    BREAK
  END_IF

LOOP

OFF -2
OFF 0 1 2              /* 緑・黄・赤LED

LOOP

```

```

*USB_MEM                                /* USB メモリーから点データを選択・読み込み・動作

ON_USB                                    /* USB MEMORY PORT ACTIVATE

DO
S_MBK 5 993                              /* "ファイルを選んで下さい" 表示
SELECT_CASE VOID
CASE SW(79000)                            /* タッチパネル ▼ button
  IF FILE_CNT<FILE_CNT_MAX THEN
    INC FILE_CNT 6
  ELSE
    FILE_CNT=0
  END_IF
  GOSUB *USB_CHA_CAP

CASE SW(79001)                            /* タッチパネル ▲ button
  IF FILE_CNT>0 THEN
    INC FILE_CNT -6
  ELSE
    FILE_CNT=FILE_CNT_MAX
  END_IF
  GOSUB *USB_CHA_CAP

CASE SW(79003)                            /* タッチパネル PLAY button
S_MBK 2 993                                /* タッチパネル "点データ名読み込み中" 表示
FN$=MBK$(994,12)
SERCH FN$ ".P2K"
IF ptr_<>0 THEN
  PR "====" FN$ "===="
  NEWP

ON_ERROR *USB_MEM_ERR /* エラー時処理
USB_PLOAD FN$ /* USB メモリから読み込み
IF err_==0 THEN /* err_は予約変数。no error なら動作する。
  S_MBK 3 993 /* タッチパネル "読み込みました"
  TIME 1000
  S_MBK 0 993 /* タッチパネル 表示無し
  BREAK
END_IF
ON_ERROR VOID /* VOID すると err_ も リセット
ELSE
  S_MBK 4 993 /* タッチパネル "点データではありません" 表示
  PRINT "invalid file"
  TIME 1000
  S_MBK 0 993 /* タッチパネル 表示無し
END_IF

CASE_ELSE
END_SELECT
WAIT IN(79000)==0
SWAP
LOOP

RETURN

*USB_MEM_ERR                            /* USB_PLOAD でエラーが起きたときの処理
S_MBK 9 8 /* これをしないと2回目以降にページが替わらない
TIME 500 /* &H40 ページの RET ボタンが「全画面に戻る」のせいか?
S_MBK &H40 8 /* エラー表示画面
WAIT MBK(10)==&H40 /* ページが替わるのを待つ
PRINT "ERROR TASK=" TASKn "STEP=" err_&&H0FFFFFFF "CODE=" err_>>24 "MSG=" ERR$(err_)
MSG$="STEP="+STR$(err_&&HFFFFFFF)+" CODE="+STR$(err_>>24)+" /* >> はシフト
S_MBK "USB MEM READ ERROR " 4000 20

```

```

S_MBK MSG$ 4020 20          /* タッチパネルにエラーが発生した行番号とコードを表示
WAIT MBK(10)<>&H40          /* [RET]ボタン押下待ち
RESUME _NEXT                /* エラー発生した行の次から再開
END

```

```

*USB_READ                    /* USBメモリからファイルデータを読む
S_MBK "                      " 994 12 /* タッチパネル 表示クリア
S_MBK 1 993                  /* タッチパネル "ファイル名取得中"
DIR 1000

```

```

/* ファイル情報の取得について
/* DIR 1000 を実行すると次のようにMBKデータエリアに情報が入ります。
/* MBK(1000):ファイル数
/* MBK(1001):トータルファイル数
/* MBK(1002):トータルディレクトリ数
/* MBK(1003):USB使用容量(MByte)
/* MBK(1004)~(1009):ファイル名1(12文字)
/* MBK(1010)~(1015):ファイル名2(12文字)
/* 以下6ワードずつファイル名

```

```

/*-----
/* 下記はダイレクトコマンドでの確認例

```

```

/* #DIR
/* 2009/10/30 09:39          741 DOREMI.P2K
/* 2009/10/30 10:37        1507 TOTORO.P2K
/* 2009/10/30 10:04        704 CHARA.P2K
/*          3個のファイル          2952 バイト
/*          0個のディレクトリ

```

```

/* A:>
/* #DIR 1000
/* #PR MBK(1000)
/* 3
/* #PR MBK(1001)
/* 3
/* #PR MBK(1002)
/* 0
/* #PR MBK(1003)
/* 1
/* #PR MBK$(1004,12)
/* DOREMI .P2K
/* #PR MBK$(1010,12)
/* TOTORO .P2K
/* #PR MBK$(1016,12)
/* CHARA .P2K
/* #

```

```

IF MBK(1000)==0 THEN        /* ファイル または USBメモリ が無い
S_MBK &H80 8                /* ※USBメモリが無くてもエラー停止しない
TIME 500
S_MBK &H40 8
WAIT MBK(10)==&H40
S_MBK "NO FILE or NO DEVICE" 4000 20
S_MBK "                      " 4020 20
WAIT MBK(10)<>&H40
RETURN 0
END_IF

```

```

FILE_CNT=0
FILE_CNT_MAX=(MBK(1000)-1)*6
GOSUB *USB_CHA_CAP
PRINT "Read Index " STR$(MBK(1000)) "Files"
S_MBK 0 993                  /* タッチパネル 文字表示無し
RETURN 1

```

```

*USB_CHA_CAP          /* 表示ファイル名変更
DST_ADR=994          /* ファイル名表示先頭アドレス
CU_ADR=2052
FOR SRC_ADR=FILE_CNT TO FILE_CNT+5 /* 6ワードコピー
  S_MBK MBK(1004+SRC_ADR) DST_ADR
  INC DST_ADR 1

  OUT MBK(1004+SRC_ADR) CU_ADR~Wrd
  INC CU_ADR 2
NEXT SRC_ADR

RETURN

*PG_TEST             /* PGA PGB MANU
S_MBK &H81 8        /* タッチパネル画面
S_MBK 5 101
S_MBK 1000 102
GET_ORIGIN=0
OLD_FA=0
OLD_FB=0
S_MBK "PUSH HOME " 140 12

DO
WAIT IN(72000)&&H1F<>0
SELECT_CASE VOID
CASE SW(72000)==1 : GOSUB *PG_TEST_PGA_CW /* タッチパネル PGA CW ボタン
CASE SW(72001)==1 : GOSUB *PG_TEST_PGA_CCW /* タッチパネル PGA CCW ボタン
CASE SW(72002)==1 : GOSUB *PG_TEST_PGB_CW /* タッチパネル PGB CW ボタン
CASE SW(72003)==1 : GOSUB *PG_TEST_PGB_CCW /* タッチパネル PGB CCW ボタン
CASE SW(72004)==1 : GOSUB *PG_TEST_PGA_HOME /* タッチパネル PGA HOME ボタン
CASE_ELSE : PRINT "?"
END_SELECT
PULSE_OUT VOID /* LED チカチカ停止
WAIT IN(72000)&&H1F==0
LOOP

*PG_TEST_PGA_CW      /* PGA CW 移動
IF GET_ORIGIN==0 THEN /* 原点復帰未完
GOSUB *PG_TEST_BUZ : RETURN
END_IF
IF MBK(101)<>OLD_FA THEN /* 前回と違うとき(都度の変更はちょっと時間がかかる)
PGA "F" MBK(101) /* タッチパネル SPEED
OLD_FA=MBK(101)
END_IF
IF PGA_POS+MBK(102)<=11000 THEN
PULSE_OUT 0 1 /* 緑LED点滅
PGA "R" MBK(102) /* タッチパネル COUNT
WAIT SW(PGA_RDY)==1 /* 移動完了確認
'PGA "C" /* "C"でその都度の取得するとちょっと時間がかかる
'PGA_POS=V_PGA
PGA_POS=PGA_POS+MBK(102) /* 現在座標
ELSE /* 可動エリア外
GOSUB *PG_TEST_BUZ
END_IF
GOSUB *PG_TEST_PGA_POS_DISP
RETURN

*PG_TEST_PGA_CCW     /* PGA CCW 移動
IF GET_ORIGIN==0 THEN /* 原点復帰未完
GOSUB *PG_TEST_BUZ : RETURN
END_IF
IF MBK(101)<>OLD_FA THEN /* 現在のタッチパネルデータと旧数値を比較
PGA "F" MBK(101) /* スピード変更

```

```

    OLD_FA=MBK(101)
END_IF
IF PGA_POS-MBK(102)>=0 THEN
    PULSE_OUT 0 1          /* 緑LED点滅
    PGA "R" MBK(102)*-1
    WAIT SW(PGA_RDY)==1
    PGA_POS=PGA_POS-MBK(102)
ELSE                          /* 可動エリア外
    GOSUB *PG_TEST_BUZ
END_IF
GOSUB *PG_TEST_PGA_POS_DISP
RETURN

*PG_TEST_PGB_CW                /* PGB CW移動
IF MBK(101)<>OLD_FB THEN
    PGB "F" MBK(101)
    OLD_FB=MBK(101)
END_IF
PULSE_OUT 1 1              /* 黄LED点滅
PGB "R" MBK(102)*-1
WAIT SW(PGB_RDY)==1
RETURN

*PG_TEST_PGB_CCW              /* PGB CCW移動
IF MBK(101)<>OLD_FB THEN
    PGB "F" MBK(101)
    OLD_FB=MBK(101)
END_IF
PULSE_OUT 1 1              /* 黄LED点滅
PGB "R" MBK(102)*-1
WAIT SW(PGB_RDY)==1
RETURN

*PG_TEST_PGA_HOME             /* PGA 原点復帰
GOSUB *PGA_HOME
PGA_POS=0                  /* PGA座標値
GET_ORIGIN=1              /* 原点復帰完了フラグ
GOSUB *PG_TEST_PGA_POS_DISP
RETURN

*PG_TEST_PGA_POS_DISP         /* PGA座標値表示
FORMAT " "                /* 文字列書式
DT$=STR$(PGA_POS)
S_MBK DT$ 140 12          /* タッチパネル文字列表示
RETURN

*PG_TEST_BUZ                  /* 警告ブザー、LEDフリック
PULSE_OUT 2 1 3          /* PORT NO, INTERVAL(*0.1SEC), COUNT
ON 3
TIME 100
OFF 3
RETURN

*PGA_HOME                     /* PGA 原点復帰(AUTO/MANU 共通)
IF SW(PGA_ORG)==1 THEN   /* 原点LS
    PGA "F" 5            /* スピード変更
    PGA "R" 1000         /* 退避移動
    WAIT SW(PGA_RDY)==1
END_IF
PGA "G" -4000            /* 4Kpps
WAIT SW(PGA_ORG)==1
OFF PGA                  /* 停止
TIME 50

```

```

PGA "H" 0          /* 座標値0セット
PGA "F" 10        /* スピード変更
RETURN

```

```

*SET_PG_ACCEL      /* 加減速テーブル作成
/* ※"A"と"F"の使い分け
/* "A"の新規加減速テーブル作成には時間がかかる。
/* (前回と同じパラメータの場合は再度作成作業は行わない。)
/* 作成したテーブルはフラッシュROMに保存するので電源を切っても消えない。
/* つまり、プログラム中で"A"は一回だけ使用、その後は変更しない。
/* スピードを変えるときは"F"を使う。
PGA "V"          /* PGA バージョン取得
PGB "V"
S_MBK V_PGA 220~Lng
PRINT "PGA, B VER" V_PGA V_PGB

```

```

PGA "D" 0          /* 2PULSE 方式
PGB "D" 0          /* 2PULSE 方式
PGA "A" 10000      /* 加減速テーブル作成 パラメータは最高速
WAIT SW(PGA_RDY)==1
PGB "A" 5000       /* 加減速テーブル作成 パラメータは最高速
WAIT SW(PGB_RDY)==1
PRINT "ACCEL SET"
RETURN

```

```

*PGA_TEST          /* PGA TEST
S_MBK &H82 8
S_MBK " PGA TEST " 140 12 /* タッチパネル 文字列表示
OFF 70000

```

```

DO
IF SW(70000)==0 THEN /* このSWはオルタネート
WAIT SW(70000)==1
GOSUB *PGA_HOME
END_IF

```

```

IF IN(25)&&HF==5 THEN
ON -1
PGA "M" 11000      /* 絶対座標移動
WAIT SW(PGA_RDY)==1
WAIT SW(-1)==0

```

```

ELSE
FOR I_ = 1 TO 11
PGA "R" 1000       /* 相対座標移動
WAIT SW(PGA_RDY)==1
NEXT I_
END_IF

```

```

IF IN(25)&&HF==5 THEN
ON -1
PGA "M" 0
WAIT SW(PGA_RDY)==1
WAIT SW(-1)==0

```

```

ELSE
PGA "M" 0
WAIT SW(PGA_RDY)==1
END_IF
TIME 500

```

```

LOOP

```

```

*PGB_TEST                               /* PGB TEST
S_MBK &H82 8
TIME 100
IF IN(25)&&HF==5 THEN
  S_MBK " PGA, B TEST " 140 12
ELSE
  S_MBK " PGB TEST " 140 12
END_IF
OFF 70000

DO
WAIT SW(70000)==1                       /* このSWはオルタネート
PGB "F" 10                               /* スピード
PGB "H" 0                                 /* 現在位置0セット

IF IN(25)&&HF==5 THEN                     /* 全面パネル DSW
  WAIT SW(-1)==1
  PGB "M" 5500
  WAIT SW(PGB_RDY)==1
  OFF -1
ELSE
  FOR I_=1 TO 5                           /* ピッチ送り
    PGB "R" 1600                           /* 2相モータ 1/2分割:400パルスで一周
    WAIT SW(PGB_RDY)==1
  NEXT I_

END_IF

IF IN(25)&&HF==5 THEN
  WAIT SW(-1)==1
  PGB "M" 0
  WAIT SW(PGB_RDY)==1
  OFF -1
ELSE
  PGB "M" 0
  WAIT SW(PGB_RDY)==1
  TIME 500
END_IF

LOOP

*PGB_DOREMI                             /* 鍵盤

OFF 73100
ON_USB

S_MBK &H86 8                             /* タッチパネル画面替え
DIR_=1                                   /* DIRECTION

DO
OUT 0 74000                              /* タッチパネル出力クリア
DO
  K_=IN(73000)&&HFF                       /* タッチパネル入力
  IF K_<>0 THEN
    BREAK
  END_IF

IF SW(73100)==1 AND REC==0 THEN
  FOR I_=2000 TO 2099
    SETP I_ 0 0 0 0
  NEXT I_
  KEY_CNT=0
  ON 1

```

```

SYSCLK=0
REC=1
END_IF
IF SW(73100)==0 AND REC==1 THEN
  USB_DEL "KEYBD. SAV" /* USB メモリ既存ファイル消去
  USB_PSAVE P(2000) 100 "KEYBD. SAV" /* USB メモリへ保存
  TIME 100 /* NEED
  OFF 1
  REC=0
END_IF

IF SW(73101)==1 AND REC==0 THEN /* PLAY ボタン
  GOSUB *PGB_DOREMI_PLAY
END_IF
SWAP

LOOP

OUT K_ 74000 /* 入力と同じ状態を出力
DIR_ =DIR_*-1 /* 一回ごとに回転方向を変える

I_SYSCLK=SYSCLK
SYSCLK=0
SELECT_CASE K_
CASE 1 : PGB "G" KEY(20)*DIR_
CASE 2 : PGB "G" KEY(30)*DIR_
CASE 4 : PGB "G" KEY(40)*DIR_
CASE 8 : PGB "G" KEY(50)*DIR_
CASE &H10 : PGB "G" KEY(60)*DIR_
CASE &H20 : PGB "G" KEY(70)*DIR_
CASE &H40 : PGB "G" KEY(80)*DIR_
CASE &H80 : PGB "G" KEY(90)*DIR_
CASE_ELSE : PR "?"
END_SELECT
WAIT IN(73000)&&HFF==0 /* タッチパネル ボタン開放待ち
OFF PGB /* PGB STOP
WAIT SW(PGB_RDY)==0

IF REC==1 THEN
  SETP 2000+KEY_CNT K_ SYSCLK DIR_ I_SYSCLK
  KEY_CNT=KEY_CNT+1
  SYSCLK=0
END_IF

LOOP

*PGB_DOREMI_PLAY /* 再生
ON 0
FOR I_=2000 TO 2099 /* 使用する範囲をクリア
  SETP I_ 0 0 0 0
NEXT I_

USB_PLOAD "KEYBD. SAV" /* USB メモリから読み込み

FOR I_=2000 TO 2099
  K_=X(I_) /* 鍵盤 NO
  T_=Y(I_) /* 回転時間
  D_=Z(I_) /* 間隔
  IF K_==0 THEN
    BREAK
  END_IF

AD_=MBK(230)/50

```



```

D_=(D_*AD_)/20
IF D_<0 THEN
  D_=0
END_IF
TIME D_
DIR_=U(I_)
OUT K_ 74000          /* 入力と同じ状態を出力
SELECT_CASE K_
  CASE 1 :   PGB "G" KEY(20)*DIR_
  CASE 2 :   PGB "G" KEY(30)*DIR_
  CASE 4 :   PGB "G" KEY(40)*DIR_
  CASE 8 :   PGB "G" KEY(50)*DIR_
  CASE &H10 : PGB "G" KEY(60)*DIR_
  CASE &H20 : PGB "G" KEY(70)*DIR_
  CASE &H40 : PGB "G" KEY(80)*DIR_
  CASE &H80 : PGB "G" KEY(90)*DIR_
  CASE_ELSE : PR "?"
END_SELECT
AD_=MBK(230)/50
T_=(T_*AD_)/20
IF T_<0 THEN
  T_=0
END_IF
TIME T_
OFF PGB
OUT 0 74000
NEXT I_

OFF 0
RETURN

```

```

*CLOCK_SET          /* タッチパネル時計 -> MPC 時計
/* GT 本体環境設定>各種設定 1
/* 時計             GT 本体時計
/* 外部転送         する
/* 出力先頭デバイス DT0
/*
/* MBK(0) nnss nn:分 ss:秒
/* MBK(1) ddhh dd:日 hh:時
/* MBK(2) yyym yy:年 mm:月
FORMAT ""
TODAY=&H20000000+MBK(2)*&H100
TODAY=TODAY+(MBK(1)/&H100)
NOW=(MBK(1)&&HFF)*&H10000+MBK(0)
SET_RTC TODAY
SET_RTC NOW
PRX TODAY NOW

RETURN

```

```

*CLOCK_TEMP_DISP   /* 時計、温度、湿度 表示
/* *注意
/* ・RS-485 と RS-232C CH1 は排他的使用。
/* ・USB-RS2 を用いてデバッグする場合、USB-RS2 の SP3-SP4 のショートジャンパー
/* を抜いてください(出荷状態では SP3-SP4 がショートされており、
/* J1 の TXD1 が RXD1 にルーブバックしてしまいます)。
/*
/* *CHINO 温湿度計 データフォーマット
/* ・送信例
/* 05 30 31 02 52 50 56 30 31 03 0D 0A
/* EQ 0 1 SX R P V 0 1 EX
/* ・応答例
/* 0 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 (文字数)

```

```

/* 06 30 31 02 41 50 56 30 31 3D 32 30 30 31 2C 20 33 2C 30 31 (HEX)
/* AK 0 1 SX A P V 0 1 = 2 0 0 1 , SP 3 , 0 1 (ASCII)
/*
/* 20 21 22 23 24 25 26 27 28 29 30 31 32 33 34 35 36 37 38 39
/* 2C 32 30 2C 34 32 2C 30 2C 30 2C 30 2C 20 20 32 32 2E 33 2C
/* , 2 0 , 4 2 , 0 , 0 , 0 , SP SP 2 2 . 3 ,
/*
/* 40 41 42 43 44 45 46 47 48 49 50
/* 30 2C 20 20 36 31 2E 37 03 0D 0A
/* 0 , SP SP 6 1 . 7 EX CR LF

```

ON\_USB

TIME 500

OFF 73100

REC=0

S\_MBK &H85 8

CH=0

ERR\_CNT1=0

ERR\_CNT2=0

S\_MBK 60 163 /\* 記録間隔 初期値

PGB "F" 10

DO

FORMAT "0000/00/00"

DD\$=HEX\$(DATE(0))

S\_MBK DD\$ 124 10 /\* タッチパネル 年月日表示

FORMAT "00:00:00"

TT\$=HEX\$(TIME(0))

OLD\_TIME=TIME(0)

S\_MBK TT\$ 116 8 /\* タッチパネル 時間表示

PRINT DD\$ TT\$

IF CH%2==0 THEN /\* 温湿度計の反応が遅いので交互に取得

GOSUB \*TEMP\_HUMI\_GET 1 135 140

TEMP1\$=TEMP\$

HUMI1\$=HUMI\$

ELSE

GOSUB \*TEMP\_HUMI\_GET 2 145 150

TEMP2\$=TEMP\$

HUMI2\$=HUMI\$

END\_IF

CH=CH+1

IF SW(73100)==1 THEN /\* タッチパネル REC ボタン

DO

IF USB(USB)==1 THEN /\* USB MEMORY 有無チェック

BREAK

END\_IF

PULSE\_OUT 2 1

S\_MBK "USB MEMORY" 124 10

S\_MBK "NONE " 116 8

WAIT SW(196)==0

WAIT SW(196)==1

PULSE\_OUT VOID

LOOP

IF REC==0 THEN /\* 記録開始

REC\_CNT=0

FOR I\_=3000 TO 6999 /\* 点データ記録エリアクリア

SETP I\_ 0 0 0 0

NEXT I\_

IF MBK(163)==10 THEN

INTV=&HF /\* 10秒間隔

```

ELSE
  INTV=&HFF          /* 60 秒間隔
END_IF
END_IF

ON 1
REC=1
IF TIME(0)&INTV==0 THEN
  REC_PNT=REC_CNT+3000
  PR "REC" REC_PNT
  X(REC_PNT)=TIME(0) /* 時間
  TEMP=VAL(TEMP1$)*10+VAL(0)
  TEMP=TEMP*&H10000 /* 上位 2 バイト CH1 温度
  HUMI=VAL(HUMI1$)*10+VAL(0) /* 下位 2 バイト CH1 湿度
  Y(REC_PNT)=TEMP+HUMI
  TEMP=VAL(TEMP2$)*10+VAL(0)
  TEMP=TEMP*&H10000 /* 上位 2 バイト CH2 温度
  HUMI=VAL(HUMI2$)*10+VAL(0) /* 下位 2 バイト CH2 湿度
  U(REC_PNT)=TEMP+HUMI
  REC_CNT=REC_CNT+1
END_IF
END_IF

IF SW(73100)==0 AND REC==1 THEN
  REC_CNT=0
  REC=0
  GOSUB *TEMP_HUMI_SAVE
  OFF 1
END_IF
S_MBK REC_CNT 162
PGB "R" AD(20)/100
WAIT SW(PGB_RDY)==1

WAIT OLD_TIME<>TIME(0)

LOOP

*TEMP_HUMI_SAVE          /* 温湿度データを USB メモリに保存
ON_ERROR *TEMP_HUMI_SAVE_ERR
FILENAME$=HEX$(DATE(0))+".CSV" /* ファイル名
USB_DEL FILENAME$      /* USB メモリ 既存ファイル削除
APPEND FILENAME$
PRINT# USB "時刻, 温度 1, 湿度 1, 温度 2, 湿度 2¥n"
FOR I_=3000 TO 6999
  IF X(I_)==0 THEN
    BREAK
  END_IF
  FORMAT "00:00:00"
  ST$=HEX$(X(I_))+", "
  FORMAT "00.0"
  ST$=ST$+STR$(Y(I_)/&H10000)+", " /* CH1 温度
  ST$=ST$+STR$(Y(I_)&HFFFF)+", " /* CH1 湿度
  ST$=ST$+STR$(U(I_)/&H10000)+", " /* CH2 温度
  ST$=ST$+STR$(U(I_)&HFFFF)+"¥n" /* CH2 湿度
  PRINT ST$
  PRINT# USB ST$
NEXT I_
PRINT "SAVED"
CLOSE USB
RETURN

*TEMP_HUMI_SAVE_ERR      /* USB でエラーが起きたときの処理
PULSE_OUT 2 1

```

```

PRINT "ERROR TASK=" TASKn "STEP=" err_&&H00FFFFFF "CODE=" err_>>24 "MSG=" ERR$(err_)
S_MBK "USB ERR " 124 10
ERNO$=STR$(err_>>24)+" "
S_MBK ERNO$ 116 8
WAIT SW(196)==1
PULSE_OUT VOID
RESUME _NEXT
END

```

```

*TEMP_HUMI_GET /* 温湿度取得
_VAR CH_NO MBK_T MBK_H
FORMAT ""
SEND$=CHR$(5) /* ENQ
SEND$=SEND$+"0"+STR$(CH_NO) /* 機器番号
SEND$=SEND$+CHR$(2) /* STX
SEND$=SEND$+"R" /* 読み出し
SEND$=SEND$+"PV" /* 測定データ
SEND$=SEND$+"01" /* データ番号
SEND$=SEND$+CHR$(3) /* ETX
SEND$=SEND$+CHR$(13)+CHR$(10)

```

```

INPUT# 1 CLR_BUF
PRINT# 1 SEND$
RCV$=""
DO
  INPUT# 1 CHR_C|1 TMOUT|2 T$
  IF T$==CHR$(&HA) OR rse_==1 THEN
    BREAK
  END_IF
  RCV$=RCV$+T$
LOOP

```

```

IF rse_==0 THEN
  STPCPY RCV$ TEMP$ 35 4
  STPCPY RCV$ HUMI$ 44 4
  S_MBK TEMP$ MBK_T 4 /* タッチパネル 温度
  S_MBK HUMI$ MBK_H 4 /* タッチパネル 湿度
  S_MBK ERR_CNT1 160
  S_MBK ERR_CNT2 161
  PRINT CH_NO TEMP$ HUMI$
ELSE
  IF CH_NO==1 THEN
    ERR_CNT1=ERR_CNT1+1
    PRINT "ERR1 " ERR_CNT1
  ELSE
    ERR_CNT2=ERR_CNT2+1
    PRINT "ERR2 " ERR_CNT2
  END_IF
  TEMP$=" ER"+STR$(rse_)
  HUMI$=""
  S_MBK TEMP$ MBK_T 4
  S_MBK HUMI$ MBK_H 4
END_IF

```

```

RETURN

```

```

*AD /* AD入力
DO
  S_MBK AD(20) 230
  TIME 100
LOOP

```

## ■USBメモリ内の演奏データ (totoro.p2k)

MPCの点データファイルの書式で保存されています。USBメモリからMPCの点データに展開して順次実行します。

*PNo*: 点番号

*X*: PGAの進行方向 1=CW 0=CCW

*Y*: KEY配列番号

*U*: 時間

*Z*: PGB回転方向

	<i>PNo</i>	<i>X</i>	<i>Y</i>	<i>U</i>	<i>Z</i>
SETP	1001	1	50	80	1
SETP	1002	1	60	80	1
SETP	1003	1	70	80	1
SETP	1004	1	75	80	1
SETP	1005	1	90	300	1
SETP	1006	1	70	100	1
SETP	1007	1	50	100	1
SETP	1008	1	10	50	1
SETP	1009	1	90	300	1
SETP	1010	1	75	100	1
SETP	1011	1	75	100	1
SETP	1012	1	60	100	1
SETP	1013	1	10	400	1
SETP	1014	1	75	300	1
SETP	1015	1	60	100	1
SETP	1016	1	40	100	1
SETP	1017	1	10	50	1
SETP	1018	1	75	300	1
SETP	1019	1	70	100	1
SETP	1020	1	70	100	1
SETP	1021	1	50	100	1
SETP	1022	1	10	500	0
SETP	1023	-1	25	300	0
SETP	1024	-1	50	300	0
SETP	1025	-1	75	300	0
SETP	1026	-1	70	300	0
SETP	1027	-1	90	80	0
SETP	1028	-1	50	300	0
SETP	1029	-1	10	50	0
SETP	1030	-1	70	70	0
SETP	1031	-1	75	70	0
SETP	1032	-1	70	70	0
SETP	1033	-1	75	70	0
SETP	1034	-1	70	70	0
SETP	1035	-1	75	70	0
SETP	1036	-1	70	70	0
SETP	1037	-1	50	70	0
SETP	1038	-1	60	800	0
SETP	1039	-1	10	10	0
SETP	1040	999	0	0	0

## ■USBメモリに保存した温湿度データ

年月日をファイル名にして、csv形式で保存します。(例 20100126.CSV)

USBメモリをPCに差し替えてEXCEL等で処理します。(下記は一分間隔で500データ)

時刻, 温度1, 湿度1, 温度2, 湿度2

08:55:00, 14.7, 29.4, 01.9, 58.8

08:56:00, 16.8, 30.1, 01.9, 63.6

08:57:00, 16.9, 28.3, 01.9, 60.4

08:58:00, 17.0, 29.0, 01.9, 63.5


(中略)

17:11:00, 25.0, 14.9, 06.8, 37.7

17:12:00, 25.0, 14.7, 06.8, 43.4

17:13:00, 24.9, 14.8, 06.6, 34.3

17:14:00, 24.9, 15.1, 06.9, 43.7



EXCELでグラフ化

以上